Ti-83 Assembly

Link port Tutorial

By Timendus

General

The link port consists of three isolated connections. The bottom one is connected to the ground of the calculator, while the voltage (with respect to the ground) of the two smaller ones can be controlled by the processor. They can be either +5 volt or +0 volt (ground). The smallest connection at the top will be referred to as the tip, the middle one is being called the ring, and the bottom one is generally being called the base. The +5 volt state of a line will be called ‘high’, the ground state of a line will be called ‘low’.

Writing to the port

If you use any romcall, both connections of the link port will be reset to their default value: both high. So be very careful with the use of romcalls in link port related programs, since they can give you terrible headaches…

If you want to manually control the link port, you have to send out a byte trough port 0 (zero). The default value for this is $D0 (or %1101.0000), which will cause both connections to turn high, if they weren’t already. To pull the lines low you have to set bits 0 and 1 of the byte. Bit 0 is the tip, while bit 1 is the ring of the link port.

So if you want to pull the ring low, but leave the tip high, you should give the following commands:

ld
a,$D2

; %11010010 (= %11010000 with bit 1 set)

out
(0),a

; Do it!

(See the schematic in appendix A for all combinations.)

Port 0 is often being referred to as “BPORT”, since BPORT is equated in Ti83asm.inc, together with the values to load into it according to the official Texas Instruments documentation (see appendix B), which I think are incorrect:

;       LINK PORT WRITE EQUATES

;

D0LD1L      .EQU         0C3H

; Don’t use these values!

D0LD1H      .EQU         0C1H

; They are incorrect!

D0HD1L      .EQU         0C2H

D0HD1H      .EQU         0C0H

BPORT       .EQU         0

It took me a few weeks to figure out that the values were the reason that all my link port related programs didn’t work (at least not on my calculator), and what the values should have been. Since then I’ve seen a few other programs use the same values, and I’ve tested my own values on every calculator I could get my hands on, and it worked on every single one of them.

Reading from the port

Reading from the link port is very much the same, except for the bits to compare. This time the correct bits are bit 2 for the tip, and bit 3 for the ring. All other bits are not interesting, so you should ‘and’ them out or simply not compare them. So if you want to find out whether the tip is being pulled low (by another calc or by someone/something else connecting it to the ground), your code should read:

in a,(0)

and 4


; %0000.0100

or a


; equal to ‘cp 0’

jr z,Tip_is_low

Tip_is_high:


etc…

Another way of doing it is by using the bit command:

in a,(0)

bit 2,a

jr z,Tip_is_low

Tip_is_high:


etc…

That’s all. It’s quite simple once you know what you’re doing, isn’t it? And again: see appendix A for all possible combinations.

Implementing it

Okay, let’s put this new knowledge to use. There are roughly two things you can do with this. First, you can use it to control hardware or read sensors (as we’ve done in our Ti controlled robot and our infrared project). Second, you can use it to exchange information with other calculators. Doing the first is relatively simple, and if you just need some software that does nothing but controlling or reading, you can use (and adapt, if you give me credit) the program “Linkport Driver” in the directory “Code examples and software” in this Zip-file. The second use of the link port is a lot more interesting.

There are again two ways to exchange bytes between calculators: the Texas Instruments way, and the good way. The routines under the [2nd] [link] menu of the calculator send bits like this:

So the received bits depend on the order in which the ports are being pulled low (This is known as the RS232 serial protocol). If you want to know all the details, I have included a very extensive set of documents, collected by Romain Lievin, describing the link port routines of most Ti calculators, in “ti_protocol.zip” in the directory “Documentation” of this Zip-file.

But there is a much faster way of communicating. By using the tip as a constant ‘clock’, you can use the ring to send the bits:

= %1001.0111 = 151

This way you can send an entire byte in just as much changes to the link port as in the above method, which sends only two bits. But then again, I have made this up myself (although there has probably been someone else before me), so if you have more confidence in Texas Instruments, feel free to use their method.

You might wonder what the use of the clock is. You need this because you have to be able to keep the bits apart. What’s the difference between %1001 and %101 if you can’t see where one zero ends and the other zero begins? That’s what the clock does: every change in the state of the clock means the beginning of the next bit.

I hope you speak fluent ASM, since we are finally going to dig into some real code! If you don’t: I’ve included “ASMguru.hlp” in the “Documentation” directory of this Zip-file for those few of you who do not already have it. It really is the best way to learn ASM currently available!

You now know how to set the lines to make them do what’s shown in the illustration above, and you know how to read the state of the lines, and translate the signals back to a byte, so you should be able to understand these routines:

Sending

Write:


ld c,a


; Store byte


ld d,1


; Create bitmask


ld e,$D1

; Init linkport value

Write_go:


ld a,c


; Retrieve byte


and d


; and with bitmask


or a


call z,Set_ring_low


call nz,Set_ring_high
; Set data line (ring) according to bit


rlc d


; rotate bitmask


ld a,e


; retrieve linkport value


out (0),a

; Set linkport


xor 1


; invert clockstate


ld e,a


; store linkport value


ld b,6

Delay_loop:


djnz Delay_loop
; Short delay


ld a,d


; bitmask back to original?


cp 1


jr nz,Write_go

; No: Next bit


ret


; Yes: Done

Set_ring_high:


res 1,e


ret

Set_ring_low:


set 1,e


ret

Receiving

Read:


ld b,0


; reset variables (b = byte,


ld d,1


; d = bitmask, e = clockstate)


in a,(0)

; Get byte and check tip


bit 2,a


call z,State_zero


call nz,State_one

Read_go:


in a,(0)

; Is clockstate changed?


bit 2,a


ld a,e


jr z,Clock_is_low


or a


jr z,Clockchanged
; Yes (High)


jr Read_go

; No

Clock_is_low:


or a


jr z,Read_go

; No  (Low)

Clockchanged:


; Yes


in a,(0)

; Get value from port


bit 2,a


call z,State_zero
; Store new clockstate


call nz,State_one


bit 3,a


call nz,Or_byte
; Or them, depending on state of ring


rlc d


ld a,d


cp 1


jr z,Stop_read

; Yes: quit


jr Read_go

; No, next bit

Or_byte:


ld a,b


or d


ld b,a


ret

State_zero:


ld e,0


ret

State_one:


ld e,1


ret

Stop_read:


ld a,b


ret
Using these routines is very simple. If you want to send over a byte, you load it in a, and you call the write routine:

ld a,164

call Write

To receive a byte, you call the read routine, that will wait for a byte being sent, and return:


call Read


ld l,a


ld h,0


call _disphl
; or whatever you want to do with a

If you don’t understand it, let me explain some of the more difficult parts. There are three things of interest in this situation: the byte, the bitmask and the clockstate. If you want to send a byte trough the link port, you have to split it up in bits somehow. That’s the use of the bitmask. It’s a byte with only one bit set, as in %0000.0001 or %0100.0000.

We start with the least significant bit (the rightmost bit), so we set the bitmask to %0000.0001. Then we take the byte we want to send, and ‘and’ it with the bitmask: if the LS bit of the byte was %1, then the result of this action will be %0000.0001. If it was %0, the result will be %0000.0000. So the next thing to do is to compare this result, and if it’s equal to zero we pull the ring low. Then we have to tell the receiving calculator that there is a new bit on the ring-line, by inverting the clockstate. Finally we rotate the bitmask one bit to the left, and we are ready to send the next bit.

The receiving calculator does nothing, until the clockstate (the state of the tip) changes. If it does, it will take a look at the state of the ring, and use it’s own bitmask to reassemble the byte at the other end of the link cable.

An intelligent addition to the receive routine might be a timeout or another failsafe, since the calculator will stop responding if it does not receive exactly eight bits. The easiest way to get your calc back (and  the way that does not involve taking out the batteries) is to put a link cable in your link port, and use a paperclip, or something a little more professional, to connect the tip to the ground of the port a few times. The calculator will then ‘receive’ the missing bits, and continue as it should.

The End

I hope you found my tutorial useful. If you have any questions, feel free to E-mail me, but please thoroughly read this document and all appendixes first. If you have an interesting link port related question I’ll add it to the last appendix, the FAQ.

If you program something using, or inspired by, my routines, or if you think this tutorial has been helpful, please give me credit. Feel free to copy this entire Zip-file or “linkport.inc” in “Code examples and software” to anyone. That way people will find out quicker that this tutorial exists (and that the official TI documentation is incorrect), and more good link port applications will appear!

ld hl,Fun_ptr

ld de,The_Matrix_progstart

ld bc,Infinity

ldir

